Introduction to Programming models

Argyris Kanellopoulos

Plant Production Systems Group, Wageningen university : argyris.kanellopoulos@wur.nl

SEAMLESS

Outline of the presentation

- Problem formulation
- Decision variables
- Objective function
- Constraints
- Solving LP problems graphically
- Exercise: the farmers problem

SEAMLESS

Problem formulation

- A farmer can choose between two main activities: growing maize or wheat
- The gross margin from growing maize is 500 €/ha
- The gross margin from growing wheat is 200 €/ha
- The total available land is 6 ha, only 4 of which are irrigated
- Maize can only grow on irrigated land

How many ha of maize and how many ha of wheat should be grown to maximize total gross margin?

SEAMLESS

Decision variables

- The unknowns
- Level of different activities (available options)
- We want to find the area of maize and the area of wheat that maximize the gross margin

X_m = the area of maize (ha)

X_w = the area of wheat (ha)

SEAMLESS

Objective function

- The function that we want to optimize
- The total gross margin as a function of the decision variables
- 1 ha of maize gives 500 €
- 1 ha of wheat gives 200 €

$$\text{T. Gross Margin} = \quad w = 500 \cdot X_m + 200 \cdot X_w$$

SEAMLESS

Constraints

- The total area of wheat and maize cannot exceed 6 ha:

$$X_m + X_w \leq 6$$

- Maize can only grow on irrigated land (=4 ha):

$$X_m \qquad \leq 4$$

SEAMLESS

Solving LP problems graphically

- Mathematical formulation of farmer's problem:

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$

SEAMLESS

Solving LP problems graphically

- Feasible area

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$



SEAMLESS

Solving LP problems graphically

- Feasible area

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$



SEAMLESS

Solving LP problems graphically

- Feasible area

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$



SEAMLESS

Solving LP problems graphically

- Objective vector points at the direction of the optimal solution

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$



SEAMLESS

Solving LP problems graphically

- Objective vector points at the direction of the optimal solution

$$\text{Max}\,\{w = 500 \cdot X_m + 200 \cdot X_w\}$$

Subject to:

$$X_m + X_w \leq 6$$
$$X_m \leq 4$$
$$X_m, X_w \geq 0$$



SEAMLESS

Solving LP problems graphically

- Iso-profit lines
- Perpendicular to obj. vector
- all points of the line have the same gross margin
- The last point of the feasible area is the optimal solution
- Optimal solutions is always a corner point



Solution:
$X_w = 2$ ha
$X_m = 4$ ha
$W^* = 2 \cdot 200 + 4 \cdot 500 = 2400$ €

Shadow price

- Shadow price is the improvement of the objective function if we relax the constraint by 1 unit
- Each constraint has a shadow price



Shadow price

- Shadow price is the improvement of the objective function if we relax the constraint by 1 unit
- Each constraint has a shadow price



Shadow price

- Shadow price is the improvement of the objective function if we relax the constraint by 1 unit
- Each constraint has a shadow price



Shadow price

- Shadow price is the improvement of the objective function if we relax the constraint by 1 unit
- Each constraint has a shadow price
- Binding constraints = constraints with positive shadow price



Exercise : Formulate an LP problem

- Define decision variables explicitly
- Determine the objective function
- Identify relevant constraints

Exercise 1 (pg 5)

Exercise 2a: farmer's problem (pg 6)

Introduction to GAMS

Argyris Kanellopoulos

Plant Production Systems Group, Wageningen university : argyris.kanellopoulos@wur.nl

SEAMLESS

Outline of the presentation

- From explicit to generic model formulation
- Farmer's problem in GAMS
- Structure of a GAMS model
- $ INCLUDE statements
- Subsets
- Exercise 2c-e
- Matrix notation of the farmer's problem
- Farmer's problem in a generic model

SEAMLESS

Explicit vs Generic formulation

- Explicit formulation
- Generic formulation

Sets: i "index for crops" $i \in (g, p, s)$

j "index for season" $j \in (Sp, Au)$

$$\text{Max } \{w = 4500 \cdot X_g + 5000 \cdot X_p + 6000 \cdot X_s\} \quad \Rightarrow \quad \text{Max } \{ w = \sum_i prof_i \cdot X_i \}$$

S.t:

$$X_g + X_p + X_s \le 10 \quad \Rightarrow \quad \sum_i X_i \le AL$$

$$\begin{aligned} 2X_g + 24X_p + 158X_s &\le 350 \\ 25X_g + 20X_p + 5X_s &\le 150 \end{aligned} \quad \Rightarrow \quad \sum_i labReq_{i,j} \cdot X_i \le AvLabr_j \quad \forall j$$

$$\begin{aligned} 2X_g &\le 10 \\ 3X_p &\le 10 \\ 4X_s &\le 10 \end{aligned} \quad \Rightarrow \quad maxCrop_i \cdot X_i \le AL \quad \forall i$$

$$X_g, X_p, X_s \ge 0$$

SEAMLESS

Exercise 2: Farmer's problem in GAMS

- Open GAMS-IDE

SEAMLESS

Exercise 2: Farmer's problem in GAMS



- Create new project

SEAMLESS

Exercise 2: Farmer's problem in GAMS

- Create new *.gms file or open an existing one
- Open farmProblem1.gms



SEAMLESS

Exercise 2: Farmer's problem in GAMS

- Define sets for crops and seasons

```
** Starting the model declar

SETS

i "index for crops" /g,p,s/
j "index for seasons"
/
sp
au
/ ;
```

Sets: i "index for crops" $i \in (g,p,s)$
j "index for season" $j \in (Sp, Au)$

Exercise 2: Farmer's problem in GAMS

- Declare parameters used in the model

```
PARAMETERS
PROF(i)          "profit from growing crop i
LABREQ(i,j)      "labour requirements for crop
AVLABR(j)        "available labour in seson j
MAXCROP(i)       "upper bound to the area of c
;

SCALARS
AL               "available land (ha)"
;
```

Exercise 2: Farmer's problem in GAMS

- Declare variables
- Positive variables x≥0

```
VARIABLES
W              "total profit and objective function value (euros)"
;

POSITIVE VARIABLES
X(i)           "activity level _ the optimal crop areas (ha)"
;
```

Exercise 2: Farmer's problem in GAMS

```
EQUATIONS
OBJECT            "the objective function"
AVLAND            "available land constraint"
LABRAV(j)         "labour availability constraints"
CRPROT(i)         "crop rotation constraints"
;

OBJECT..          sum(i,PROF(i)*X(i))          =E= W         ;

AVLAND..          sum(i,X(i))                  =L= AL        ;

LABRAV(j)..       sum(i,LABREQ(i,j)*X(i))      =L= AVLABR(j) ;

CRPROT(i)..       MAXCROP(i)*X(i)              =L= AL ;
```

$w = \sum_i prof_i \cdot X_i$

$\sum_i X_i \leq AL$

$\sum_i labReq_{i,j} \cdot X_i \leq AvLabr_j \quad \forall j$

$maxCrop_i \cdot X_i \leq AL \quad \forall i$

- Declare Objective function & constraints
- Define constraints

Exercise 2: Farmer's problem in GAMS

```
MODEL simpleFarm "the simple farm model in gams"
/
OBJECT
AVLAND
LABRAV
CRPROT
/ ;
```

- A Model is a set of equations
- You can switch on and off equations only by including and excluding them from the set

Exercise 2: Farmer's problem in GAMS

```
** Starting the data initialization section

PARAMETER PROF(i)          "profit from growing crop i (euros/ha)"
/
g   4500
p   5000
s   6000
/;

PARAMETER LABREQ(i,j)      "labour requirements for crop i in season j (hours/ha)"
/
g.sp  2
g.au  25
p.sp  24
p.au  20
s.sp  158
s.au  5
/;

PARAMETER AVLABR(j)        "available labour in seson j (hours)"
/
sp  350
au  150
/;

PARAMETER MAXCROP(i)       "upper bound to the area of crop i (ha)"
/
g  2
p  3
s  4
/;


AL = 10 ;


** Ending the data initialization section
```

Exercise 1: Formulate an LP problem

```
** Ending the data initialization section

       Model name   Problem type
SOLVE  simpleFarm USING LP MAXIMIZING W ;

DISPLAY W.L, X.L, PROF, LABRAV.M; Shadow price of constraint
     Level of variable
```

- Solving the model
- Reporting specific results and data
- Run the model by pressing F9

SEAMLESS

Exercise 2: Farmer's problem in GAMS

C:\argyrisKanellopoulos\presentations\FSSIM workshop (06-091211)\day1\Exercise1_farm_problem\farmProblem1.lst

farmProblem1.gms | farmProblem1.lst

```
Compilation
 Equation Listing   SOLVE sim
 Equation
 Column Listing     SOLVE sim
 Column
 Model Statistics   SOLVE sim
 Solution Report    SOLVE sim
 SolEQU
 SolVAR
   W
   X
Execution
 Display
```

```
                     LOWER     LEVEL      UPPER    MARGINAL

---- VAR W           -INF    40188.240    +INF       .

  W  total profit and objective function value (euros)

---- VAR X  activity level _ the optimal crop areas (ha)

        LOWER     LEVEL     UPPER    MARGINAL

g         .       2.999     +INF       .
p         .       3.333     +INF       .
s         .       1.671     +INF       .

**** REPORT SUMMARY :        0     NONOPT
                             0 INFEASIBLE
                             0  UNBOUNDED
```

SEAMLESS

Exercise 2: Farmer's problem in GAMS

C:\argyrisKanellopoulos\presentations\FSSIM workshop (06-091211)\day1\Exercise1_farm_problem\farmProblem1.lst

farmProblem1.gms | farmProblem1.lst

```
Compilation
 Equation Listing   SOLVE sim
 Equation
 Column Listing     SOLVE sim
 Column
 Model Statistics   SOLVE sim
 Solution Report    SOLVE sim
 SolEQU
   OBJECT
   AVLAND
   LABRAV
   CRPROT
 SolVAR
   W
   X
Execution
 Display
```

```
                    LOWER     LEVEL     UPPER    MARGINAL

---- EQU OBJECT       .         .         .       -1.000
---- EQU AVLAND     -INF      8.003    10.000       .

  OBJECT  the objective function
  AVLAND  available land constraint

---- EQU LABRAV  labour availability constraints

      LOWER     LEVEL     UPPER    MARGINAL

sp    -INF    350.000   350.000    32.360
au    -INF    150.000   150.000   177.411

---- EQU CRPROT  crop rotation constraints

      LOWER     LEVEL     UPPER    MARGINAL

g     -INF      2.999     5.000      .
p     -INF      3.333     3.333   675.127
s     -INF      1.671     2.500      .
```

- Marginal = shadow price = increase of obj. value if constraint is relaxed by 1 unit

SEAMLESS

Structure of a GAMS model

- Title and options
- Set declaration
- Parameter declaration
- Variable declaration
- Equation declaration
- Equation definition
- Model definition
- Data Initialization
- SOLVE statements
- DISPLAY reporting

```
AL = 5 ;

SOLVE  simpleFarm USING LP MAXIMIZING W ;

DISPLAY W.L, X.L, PROF, LABRAV.M;
```

SEAMLESS

$ INCLUDE statement

Project_dir\dataInitialization.gms

```
PARAMETER PROF(i)        "profit from growing crop i (euros/ha)"
/
g   4500
p   5000
s   6000
/;

PARAMETER LABREQ(i,j)    "labour requirements for crop i in season j (hours/ha)"
/
g.sp  2
g.au  25
p.sp  24
p.au  20
s.sp  158
s.au  5
/;

PARAMETER AVLABR(j)      "available labour in seson j (hours)"
/
sp  350
au  150
/;

PARAMETER MAXCROP(i)     "upper bound to the area of crop i (ha)"
/
g  2
p  3
s  4
/;


AL = 10 ;

** Ending the data initialization section
```

```
** Starting the data initialization section
$ INCLUDE dataInitialization.gms
** Ending the data initialization section
```

SEAMLESS

$ INCLUDE statement

```
PARAMETER LABREQ(i,j)     "labour requirements for crop i in season j (hours/ha)'
/
g.sp  2
g.au  25
p.sp  24
p.au  20
s.sp  158
s.au  5
/;
```

Project_dir\data\LABREQ.inc

```
PARAMETER LABREQ(i,j)     "labour re
/
$ INCLUDE data\LABREQ.inc
/;
```

SEAMLESS

Subsets

```
SETS
i "Cities" /AMSTERDAM,BERLIN,LONDON,UTRECT,ROME,PARIS/

j(i) "Dutch cities"
;

j(i)            = NO ;
j("AMSTERDAM")  = YES;
j("UTRECT")     = YES;
```

```
BOUND_COFFEE_SHOPS(i)$j(i).. X(i) =L= allowedCoffeeShops(i);
NR_COFFEE_SHOPS..            sum(i$j(i),X(i)) =E= W  ;
* maximize W :))
```

SEAMLESS

Exercise 1: Formulate an LP problem

Exercise 2c-e: farmer's problem (page 6)

SEAMLESS

Matrix notation of the farmer's problem

- Explicit formulation

$$\text{Max } \{w = 4500 \cdot x_g + 5000 \cdot x_p + 6000 \cdot x_s$$

S.t:

$$\begin{aligned} x_g + x_p + x_s &\le 10 \\ 2\,x_g + 24\,x_p + 158\,x_s &\le 350 \\ 25\,x_g + 20\,x_p + 5\,x_s &\le 150 \\ 2\,x_g &\le 10 \\ 3\,x_p &\le 10 \\ 4\,x_s &\le 10 \\ x_g, x_p, x_s &\ge 0 \end{aligned}$$

- matrix formulation

$$\text{Max } \begin{bmatrix} 4500 & 5000 & 6000 \end{bmatrix} \times \begin{bmatrix} x_g \\ x_p \\ x_s \end{bmatrix}$$

S.t:

$$\begin{bmatrix} 1 & 1 & 1 \\ 2 & 24 & 158 \\ 25 & 20 & 5 \\ 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 4 \end{bmatrix} \times \begin{bmatrix} x_g \\ x_p \\ x_s \end{bmatrix} \le \begin{bmatrix} 10 \\ 350 \\ 150 \\ 10 \\ 10 \\ 10 \end{bmatrix}$$

$$x_g, x_p, x_s \ge 0$$

SEAMLESS

Matrix notation of the farmer's problem

$$\text{Max } \begin{bmatrix} 4500 & 5000 & 6000 \end{bmatrix} \times \begin{bmatrix} x_g \\ x_p \\ x_s \end{bmatrix}$$

S.t:

$$\begin{bmatrix} 1 & 1 & 1 \\ 2 & 24 & 158 \\ 25 & 20 & 5 \\ 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 4 \end{bmatrix} \times \begin{bmatrix} x_g \\ x_p \\ x_s \end{bmatrix} \le \begin{bmatrix} 10 \\ 350 \\ 150 \\ 10 \\ 10 \\ 10 \end{bmatrix}$$

$$x_g, x_p, x_s \ge 0$$

$$\max \{w = c'x\}$$

s.t:

$$Ax \le b$$

$$x \ge 0$$

SEAMLESS

Matrix notation of the farmer's problem

```
SETS
C "coefficients rows of A matrix"
/
C_LAND
C_LABR_SP
C_LABR_AU
C_LEV_G
C_LEV_P
C_LEV_S
/

R "resources rows of b vector"
/
R_ALAND
R_ALABR_SP
R_ALABR_AU
R_LEV_G
R_LEV_P
R_LEV_S
/

L_CR(C,R) "constraints -> combinations of C and R
;

L_CR("C_LABR_SP","R_ALABR_SP") = YES ;
```

SEAMLESS

Exercise 2: Farmer's problem in a generic model

- Open the project ...day1\ex2_generic\generic.gps
- Open existing file experiment.gms
- Notice the include statements
- Open the included files
- Solve exercise 2 by only adapting data files in the "data" directory.

Exercise 2c-e: farmer's problem (page 6)

SEAMLESS

Standard PMP and extensions

Argyris Kanellopoulos

Plant Production Systems Group, Wageningen university : argyris.kanellopoulos@wur.nl

SEAMLESS

Outline of the presentation

- Overspecialization of LP solutions
- Standard PMP
- Exercise on standard PMP
- Extensions and applications

SEAMLESS

Overspecialization of LP solution

Max $\{w = 500 \cdot X_m + 200 \cdot X_w\}$

Subject to:

$X_m + X_w \leq 6$

$X_m \leq 4$

$X_m,\ X_w \geq 0$



Solution:

X_w = 2 ha

X_m = 4 ha

$W^* = 4 \cdot 500 + 2 \cdot 200 = 2400$ €

SEAMLESS

Overspecialization of LP solution

Problem:

- LP solution always a corner point
- Number of selected activities ≤ number of binding constraints
- Not easy to identify all farm specific constraints that result in the observed activity levels
- Function of variable costs is usually non linear
- Not enough degrees of freedom for estimation

Solution:

- Recover the unknown coefficients of a non-linear cost function, using PMP

SEAMLESS

Overspecialization of LP solution

- What if $x^o = (x_w^o, x_m^o) = (3\ ,3)$?



SEAMLESS

Overspecialization of LP solution

- Using hard calibration constraints



SEAMLESS

Overspecialization of LP solution

- Non linear objective function



SEAMLESS

Standard PMP: step 1

- Fix simulated activity levels to the observed activity levels + ε
- Use (temporary) calibration constraints



SEAMLESS

Standard PMP: step 1

- Fix simulated activity levels to the observed activity levels + ε
- Use (temporary) calibration constraints



SEAMLESS

Standard PMP: step 1

- Fix simulated activity levels to the observed activity levels + ε
- Use (temporary) calibration constraints
- Calculate shadow prices of calibration constraints



Shadow prices:

$\lambda_w = 0$

$\lambda_m = 300$

SEAMLESS

Standard PMP: step 2

$$\text{Max } \{w = 500 \cdot X_m + 200 \cdot X_w\}$$

S.t:

$$\begin{aligned} X_m + X_w &\leq 6 \\ X_m &\leq 4 \\ X_m &\leq 3 + 0.001 \\ X_w &\leq 3 + 0.001 \\ X_m, X_w &\geq 0 \end{aligned}$$

$$\text{Max } \{w = 500 \cdot X_m + 200 \cdot X_w - 0.5 \cdot q_{mm} \cdot X^2_m - 0.5 \cdot q_{ww} \cdot X^2_w - 0.5 \cdot q_{wm} \cdot X_w \cdot X_m - 0.5 \cdot q_{mw} \cdot X_m \cdot X_w$$

S.t:

$$\begin{aligned} X_m + X_w &\leq 6 \\ X_m &\leq 4 \\ X_m, X_w &\geq 0 \end{aligned}$$

- Exact calibration if:

$$\left.\begin{aligned} Prof_m - \lambda_m &= Prof_m - q_{mm} \cdot X^o_m - q_{wm} \cdot X^o_w \\ Prof_w - \lambda_w &= Prof_w - q_{ww} \cdot X^o_w - q_{wm} \cdot X^o_m \end{aligned}\right\} \begin{aligned} \lambda_m &= q_{mm} \cdot X^o_m + q_{mw} \cdot X^o_w \\ \lambda_w &= q_{ww} \cdot X^o_w + q_{wm} \cdot X^o_m \end{aligned}$$

- Assumptions in standard PMP:

$q_{wm}, q_{mw} = 0,$

$q_{mm} \cdot X^o_m = \lambda_m \Rightarrow q_{mm} = \lambda_m / X^o_m$

$q_{ww} \cdot X^o_w = \lambda_w \Rightarrow q_{ww} = \lambda_w / X^o_w$

SEAMLESS

Gross margins in LP



SEAMLESS

Gross margins Std. PMP

SEAMLESS

Exercise 1: Formulate an LP problem

$$\max \{w = c'x\}$$
s.t:
$$Ax \leq b$$
$$x \geq 0$$

$$\max \{w = c'x\}$$
s.t:
$$Ax \leq b$$
$$x \leq x^o + \varepsilon$$
$$x \geq 0$$

$$q_{ij} = 0$$
$$q_{ii} = \lambda_i / \chi_i^o$$

$$\max \left\{w = c'x - \frac{1}{2}x'Qx\right\}$$
s.t:
$$Ax \leq b$$
$$x \geq 0$$

Max $\{w = 500 \cdot X_m + 200 \cdot X_w\}$
S.t:
$X_m + X_w \leq 6$
$X_m \leq 4$
$X_m, X_w \geq 0$

Max $\{w = 500 \cdot X_m + 200 \cdot X_w\}$
S.t:
$X_m + X_w \leq 6$
$X_m \leq 4$
$X_m \leq 3 + 0.001$
$X_w \leq 3 + 0.001$
$X_m, X_w \geq 0$

$$q_{mw}\ q_{wm} = 0$$
$$q_m = \frac{300}{3} = 100$$
$$q_w = \frac{0}{3} = 0$$

Max $\{w = 500 \cdot X_m + 200 \cdot X_w - 0.5 \cdot 100 \cdot X^2_m\}$
S.t:
$X_m + X_w \leq 6$
$X_m \leq 4$
$X_m, X_w \geq 0$

SEAMLESS

PMP variants

$$Prof_m - \lambda_m = Prof_m - q_{mm} \cdot X^o_m - q_{wm} \cdot X^o_w$$
$$Prof_w - \lambda_w = Prof_w - q_{ww} \cdot X^o_w - q_{wm} \cdot X^o_m$$

- Heckelei, 2002. Review of PMP variants
- Rohm and Dabbert, 2003 -> substitution between management -> require observations at crop/management level
- Kanellopoulos et al., 2010 -> use information on elasticities or ex-post experiments -> requires data on elasticities or multi-year historical data.
- Paris and Howit, 1998 and Heckelei and wolf, 2003 -> multiple year observation and Generalized Maximum Entropy

SEAMLESS

Exercise 3: PMP calibration

Exercise 3: Calibrating the farmer's problem (pg 6)

SEAMLESS

The Farm System SIMulator V2.0:

a generic bio-economic farm model

Argyris Kanellopoulos

Plant Production Systems Group, Wageningen university : argyris.kanellopoulos@wur.nl

SEAMLESS

Outline of the presentation

- Mathematical model
- Conceptualization of activities
- Data structure
- Important files of FSSIM V2.0
- Exercise: the farmer's problem in FSSIM

SEAMLESS

Mathematical formulation

$$\max \{ w = c'x - NLP - INT \}$$
$$s.t.:$$
$$Ax \leq b$$
$$x \geq 0$$

- Bio-economic farm model
- Maximize an objective function subject to resource, rotational and policy constraints
- Non-linear mixed integer model
- Basic model is an LP model (i.e. NLP, INT =0)

Conceptualization of activities



Example of activities: SEAMLESS



Example of activities: AgriadaptNL (DEA)



Data structure



farmData.mdb



Experiment.gms

- Title and options
- Set declaration
- Parameter declaration
- Variable declaration
- Equation declaration
- Equation definition
- Data Initialization (data for all simulations and farms)
- Model definitions
- Load farm and simulation specific data
- SOLVE statements
- DISPLAY reporting

For all

```
C:\argyrisKanellopoulos\presentations\FSSIM workshop (06-091211)\da
experiment.gms
$ TITLE  FSSIM FOR WORKSHOP

OPTIONS LP = CONOPT3, NLP = CONOPT3, MINLP = SBB ;

*** Main component of FSSIM-MP ***
$ INCLUDE farmModel.gms
$ INCLUDE dataInitialization.gms

*** plugIns used in Experiment ***
$ INCLUDE plugIns.gms

*** model declarations ***
$ INCLUDE modelDeclarations.gms

*** simulations of the experiment ***

LOOP(F,
* Simulation of the current situation only current
$ INCLUDE simulations\SIM1\farmData.gms
SOLVE farmModelLP using LP maximizing OBJ  ;
$ INCLUDE simulations\SIM1\reportResults.gms
) ;

DISPLAY ACTIVITY_LVL, CONSTR_SHAD, INDICATORS, OBJ_

*** Aggregate results to 2 different farm types
$ INCLUDE resultAggregation.gms
```

SEAMLESS

farmModel.gms

```
OBJECTIVE.. OBJ  =E=    sum(C$OB(C),W(C)*T_COEF(C))
                   -    QUADR_costs
                   -    INTEG
;
TOTALS(C)..           sum(N,COEF(C,N)* X(N))  =E= T_COEF(C)
;
QUADRAT_NULL ..                    QUADR_costs =E= 0
;
INTEG_NULL ..                            INTEG =E= 0
;
AMAT_L(C,R)$L_CR(C,R)..               T_COEF(C)  =L= RESOURCE(R)
;
AMAT_G(C,R)$G_CR(C,R)..               T_COEF(C)  =G= RESOURCE(R)
;
NON_FN_NULL..        sum((N)$(NOT FN(N)),X(N))  =E= 0
;
```

e.g. w(gm)*gm + w(risk)*Risk

1 *gm + 0.65 *Risk

e.g. Labour/ha of activity N -> Total labor Requirements

e.g. total Labor req. ≤ Available labour

SEAMLESS

dataInitialization.gms

```
*    <Initializing set N>
     SET N "" /
$    include data/N.inc
     /;
*    </Initializing set N>


*    <Initializing set R>
     SET R "" /
$    include data/R.inc
     /;
*    </Initializing set R>


*    <Initializing set C>
     SET C "" /
$    include data/C.inc
     /;
*    </Initializing set C>
```

```
PARAMETER WEIGHT_ALL(S,F,C) "the available resou
     /
$    include data/WEIGHT_all.inc
     / ;
*    </Initializing parameter W>

*    <Initializing parameter COEF>
      COEF(C,N) = 0 ;

PARAMETER COEF_ALL(S,F,N,C) "Data for the input
     /
$    include data/COEF_all.inc
     /;
COEF_ALL(S,F,N,C)$(abs(COEF_ALL(S,F,N,C))<=0.000
     </Initializing parameter COEF>


*    <Initializing parameter RESOURCE>
      RESOURCE(R) = 0;

PARAMETER RESOURCE_ALL(S,F,R) "the available re
     /
$    include data/resource_all.inc
     / ;
*    </Initializing parameter RESOURCE>
```

SEAMLESS

dataInitialization.gms

```
$setlocal DataRepository ..\data
$setlocal commandfile commands.txt
$onecho > %commandfile%
I=%system.fp%farmData.mdb

Q1="select * FROM SET_F;"
O1=data\F.inc

Q2="select * FROM SET_N;"
O2=data\N.inc
```

: : : : : : :

```
Q11="select * FROM P_SF_GCR;"
O11=data\P_SF_GCR.inc

Q12="select * FROM P_SF_OB;"
O12=data\P_SF_OB.inc

$offecho

$call =mdb2gms @%commandfile%
```

SEAMLESS

\simulations\SIM1\farmData.gms

LOOP (F,

SIM("SIM1") = YES

```
*    <Initializing parameter COEF>
      COEF(C,N) = 0;
      COEF(C,N) = sum(S$SIM(S),COEF_ALL(S,F,N,C))   ;
*    </Initializing parameter COEF>


*    <Initializing parameter RESOURCE>
      RESOURCE(R) = 0;
      RESOURCE(R) = sum(S$SIM(S),RESOURCE_ALL(S,F,R)) ;
*    </Initializing parameter RESOURCE>
```

Initialized in dataInitialization.gms

SOLVE farmModelLP using LP maximizing OBJ;

);

SEAMLESS

Output: IO.xls

- Reporting results per simulation
- Individual farm simulations and average farm of specific farm type
- Main information that is reported:
  - Indicators (T_COEF(C))
  - Activity levels (X.L(N))
  - Shadow prices of constraints
  - Objective function values (OBJ.L)
  - Model and Solver status

SEAMLESS

Exercise 3: PMP calibration

Exercise 4:The farmers problem in FSSIM V2.0 (pg 7)

SEAMLESS